Intégration par parties

Jean-Michel Sarlat
13 avril 2003

Ce document a été généré par la ***fabrique de Syracuse*** *à partir d'un fichier texte contenant des spécifications de contenu et des commandes d'interfaçage avec* ***Maxima****. Le traitement de ce fichier a donné lieu, dans le même temps que ce document, à la création d'une page* html.
URL : http ://melusine.eu.org/syracuse/maxima/

L'*intégration par parties* est une technique bien utile pour calculer des primitives. **Maxima** ne propose pas de procédure spécifique pour l'invoquer, il est bien sûr possible de la mettre en scène en créant des macros adaptées.

```
ipp_a(u,v,x) := block([U],
                   U:integrate(u,x),
                   'integrate(u*v,x) = U*v -'integrate(U*diff(v,x),x));
```

Remarque – La commande `integrate`(*expression*,*variable*) permet de calculer une primitive de l'*expression* selon la *variable*. Lorqu'elle est précédée d'une apostrophe **'** elle est rendue inerte dans le sens où elle n'est pas évaluée à priori, cela permet de disposer de son affichage en tant qu'intégrale.

1 Premiers exemples

maxima >>

```
(C2) load("ipp.mc")$
(C3) load("integration.mc")$
(C4) primitive(asin(x)^2,x);
```

(D4) $$\int \operatorname{Arcsin}^2 x \, dx = x \operatorname{Arcsin}^2 x + 2\sqrt{1-x^2}\operatorname{Arcsin} x - 2x$$

```
(C5) ipp_a(1,asin(x)^2,x);
```

(D5) $$\int \operatorname{Arcsin}^2 x \, dx = x \operatorname{Arcsin}^2 x - 2\int \frac{x \operatorname{Arcsin} x}{\sqrt{1-x^2}} \, dx$$

```
(C6) ipp_a(x/sqrt(1-x^2),asin(x),x);
```

(D6)
$$\int \frac{x\ \mathrm{Arcsin}\,x}{\sqrt{1-x^2}}\,dx = x - \sqrt{1-x^2}\ \mathrm{Arcsin}\,x$$

L'intégrale qui aurait dû apparaître dans le second membre n'est pas au rendez-vous ! C'est normal la variable ne figure plus dans l'intégrande (il vaut `1`), l'*expression* sur laquellle porte l'intégration est constante vis à vis de la variable d'intégration, le calcul est *effectif*. De la même façon **maxima** effectuera la factorisation par une constante (si cette constante factorise l'intégrande) dans la forme inerte de l'intégrale.

```
(C7) ipp_a(1,log(t)^2,t);
```

(D7)
$$\int \ln^2 t\,dt = t\ \ln^2 t - 2\int \ln t\,dt$$

```
(C8) ipp_a(1,log(t),t);
```

(D8)
$$\int \ln t\,dt = t\ \ln t - t$$

2 À la recherche d'une relation de récurrence

maxima >>

```
(C2) load("ipp.mc")$
(C3) f[n](x) := 1/(x^2+1)^n;
```

(D3)
$$f_n(x) = \frac{1}{(x^2+1)^n}$$

```
(C4) I[n](x) := integrate(f[n](x),x);
```

(D4)
$$I_n(x) = \int \frac{1}{(x^2+1)^n}\,dx$$

```
(C5) A:ipp_a(1,f[n](x),x);
```

(D5)
$$\int \frac{1}{(x^2+1)^n}\,dx = 2\,n\int x^2\ \left(x^2+1\right)^{-n-1}\,dx + \frac{x}{(x^2+1)^n}$$

```
(C6) block(B:'(f[n](x)-f[n+1](x)),B = ev(B,radcan,nouns));
```

(D6) $$f_n(x) - f_{n+1}(x) = x^2\ \left(x^2+1\right)^{-n-1}$$

```
(C7) C:'(I[n](x)-I[n+1](x)) = integrate(rhs(%),x);
```

(D7) $$I_n(x) - I_{n+1}(x) = \int x^2\ \left(x^2+1\right)^{-n-1}\,dx$$

```
(C8) subst(lhs(%),rhs(%),subst('(I[n](x)),I[n](x),A));
```

(D8) $$I_n(x) = 2\,n\ \left(I_n(x) - I_{n+1}(x)\right) + \frac{x}{\left(x^2+1\right)^n}$$

```
(C9) solve(%,'(I[n+1](x)));
```

(D9) $$\left[I_{n+1}(x) = \frac{(2\,n-1)\ \left(x^2+1\right)^n\ I_n(x) + x}{2\,n\ \left(x^2+1\right)^n}\right]$$

```
(C10) distrib(part(%,1));
```

(D10) $$I_{n+1}(x) = \frac{(2\,n-1)\ I_n(x)}{2\,n} + \frac{x}{2\,n\ \left(x^2+1\right)^n}$$

Voici la relation de récurrence recherchée. On peut donc, de proche en proche, calculer les primitives `I[n](x)` ... ce que **maxima** sait faire de toute façon.

```
(C11) load("integration.mc")$
(C12) primitive(f[2](x),x);
```

(D12) $$\int \frac{1}{\left(x^2+1\right)^2}\,dx = \frac{\operatorname{Arctan} x}{2} + \frac{x}{2\,x^2+2}$$

```
(C13) primitive(f[4](x),x);
```

(D13) $$\int \frac{1}{\left(x^2+1\right)^4}\,dx = \frac{5\ \operatorname{Arctan} x}{16} + \frac{15\,x^5+40\,x^3+33\,x}{48\,x^6+144\,x^4+144\,x^2+48}$$

```
(C14) primitive(f[5](x),x);
```

(D14) $$\int \frac{1}{\left(x^2+1\right)^5}\,dx = \frac{35\ \operatorname{Arctan} x}{128} + \frac{105\,x^7+385\,x^5+511\,x^3+279\,x}{384\,x^8+1536\,x^6+2304\,x^4+1536\,x^2+384}$$